
1 – Introduction


1.1 – Why learn VBScript?

Learning VBScript lets you:


	
automate tasks on Windows



	
work on Classic ASP websites



	
master other programming languages faster





Automate tasks on Windows systems

Here are just some of the things you can automate:


	
moving/renaming files



	
processing file contents



	
running and controlling other applications





This book will introduce you to all three.

Work on Classic ASP websites

Classic ASP is VBScript.

It has some adaptations for website use, but it’s the same language.

Classic ASP is still in use. Companies have legacy systems and need people to maintain them. You could be one of those people, now or in the future. If you are, knowing VBScript will give you the edge.

Master other programming languages faster

There’s a whole family of VB languages:


	
VB.Net



	
VBA (for Word, Excel, Access, etc)



	
VB





They share a lot with each other. Knowing VBScript will help you with the others.


1.2 – Why learn from this book?

This book will show you:


	
how to get started with VBScript



	
some practical uses of scripting



	
how to keep scripts readable, reliable, and efficient





It will save you:


	
time & typing





This book is based around a series of example scripts. Each script introduces new parts of the language and ways to use it. This is followed by an explanation, to say how the script works, what’s new, and why it matters. Problem-solving exercises are included, so you can practise as you learn, and the answers are provided on the page after.

The book starts with the basics and builds up to more advanced topics step-by-step. Each new topic is covered in a concise, focused way.

Try it out.


1.3 – Author’s Note

I hope you find this book a good way to get started with VB Script.

If anything about the book doesn’t live up to your expectations, you can let me know at:


authordarmstrong@gmail.com



2 – Get started


2.1 – Set-up instructions

Create a new folder on your computer.

Call it "VBScripts".

You can create, keep and run the scripts in this book inside this folder.

Open that folder.

In the file explorer window, on the menu bar, click "View".

On the View menu, tick the box for "Show file extensions".

This lets you not only see file extensions, but edit them, which makes it easy to create VBScript files.

You are now ready to start creating VBScripts.


2.2 – Make a script file

Example


	
Right-click the empty space in the file explorer, where the files normally are.



	
A context menu will appear. On the menu, go to "New".



	
A submenu will appear. On the submenu, click "text document".



	
A file will appear in the folder, as "New Text Document.txt". The file name, the part before the "." symbol, will be highlighted. Type "Main" to overwrite the name.



	
The ".txt", the extension, is also editable. Change it to ".vbs".



	
Press enter to confirm the change.



	
A pop-up message will appear warning you about changing the file extension. Click "yes" to proceed.





You will see the image of the file icon change.

Explanation

That’s all you need to do to get started with VBScript. No installation needed.

The ability to run VBScript is built into Windows. The default file extension for VBScript files is ".vbs". As soon as you make ".vbs" the file extension, Windows recognises the file as a VBScript.

If you double-click on the file now, Windows will run it as a script. The script won’t do anything, because we didn’t put any code in it yet, but Windows knows to treat it as a computer program, rather than, for example, open it in Notepad as a text file.

Exercise 1

Make another file the same way. Call it "Test.vbs".

We will use these two script files, Main and Test, over the next two chapters. Scripts tend to build on the scripts before them, so we’ll edit the files as we go, to save on typing.


Answer 1

No answer – this is not a problem-solving exercise.


2.3 – Make a script display a message

Example

Right-click on "Main.vbs" and select "edit" on the context menu. This opens the script file in Notepad, so you can start writing code inside.

Type the following:

MsgBox "Hello!"

Save and close the file.

Double-click the file to run it.

You should see a pop-up window appear, with the text "Hello!" as the message.

Explanation

The "MsgBox", short for "message box", is a simple way to make a script provide information. That information could be for the user, or it could be for you, as you develop and test the script.

The quotation marks identify the start and end of the text for the message. This text is known as a string. It’s necessary to mark out strings in this way, so the computer doesn’t try to run the message as a command of the VBScript language.

Exercise 2

Make the script say "Hi!" instead.


Answer 2

MsgBox "Hi!"


2.4 – Understand error messages

Example

Edit into the Main script file again.

Change the script to:

MsgBo "Hi!"

That is, delete the "x" from MsgBox

Run the script.

A pop-up will appear telling you an error has occurred. It will tell you which line in the file has the error, which character of that line, and what the error is.

In this case, it’s a "Type mismatch" error, on the first character of the first line.

Explanation

VBScript doesn’t recognise "MsgBo" as valid code, and flags it as an error.

Exercise 3

First, fix the error you just created.

Next, find out what error you get if you delete the closing quotation mark.


Answer 3

If you run:

MsgBox "Hi!

You get "Unterminated string constant".

That is, the string "Hello!" isn’t ended (terminated) in the code.

As you work through the exercises in this book, take the opportunity to learn from any error messages you get. This will be easier to do while the scripts are still small. You can even make a few typos deliberately, to see what error messages they generate. Then you’ll be better prepared to deal with any unexpected errors later.


2.5 – Make a script wait

Example

Change the Main script to:

WScript.Sleep(1000)

MsgBox "Hi!"

Run the script again.

This time, it should take a second for the message to appear.

Explanation

Sleep lets you pause the program for a specified time. That time is measured in milliseconds (ms). In the example, by pausing for 1000 milliseconds, we made the script pause for 1 second, before showing its pop-up message.

Sleep will come in useful later when we write scripts to interact with other programs, and need to make our scripts wait for them, e.g. while they load.

Notice that we didn’t have to put quotes around the number. Numbers are recognised automatically in VBScript. Numbers without decimal places are called integers.

Also, notice how we passed values to both MsgBox and Sleep, but we put brackets around one value, but only a space before the other. MsgBox and Sleep are types of method (we’ll go into more detail on methods later). Methods accept values in both of these ways.

We could just as well write:

WScript.Sleep 1000

MsgBox("Hi!")

There are some conditions around when we can use each way. We’ll cover those as they come up.

Exercise 4

Make the script say "Hi!", wait two seconds, and say "Hi again!".


Answer 4

MsgBox "Hi!"

WScript.Sleep(2000)

MsgBox "Hi again!"


2.6 – Lay out your code

Example

Change the Main script to:

wscript.sleep ( 1000 )

msgbox "hi !"

Run the script again.

Explanation

We changed the capitals to lower case, added spacing and blank lines, and the script still worked.

Only changes in the message string had any effect. Outside of text strings, VBScript ignores:


	
the case of text



	
extra spaces (groups are treated the same as single spaces)



	
extra lines-breaks (groups are treated the same as single line-breaks)





This means you can use all of these, however you like. Use them to lay out your code however you prefer. This becomes useful with longer scripts, as it helps keep them readable.

It also means that, although this book uses capital letters, to make the scripts easier to read, you don’t have to. If it’s faster or easier for you to type all in lower case, you can do that. Your scripts will still work.


3 – Learn the basics


3.1 – Use a constant

Example

Change Main.vbs to:

Const Greeting = "Hi"

MsgBox Greeting

Run it and see if it works.

Explanation

Const is short for "constant". We use "Const" to declare the word after it as a constant. In this case, we declare the word "Greeting".

In code, a constant holds a value: the string "Hi", for example. We assign a value to the constant using "=". Other parts of the code can then use that value by referring to the constant’s name. In this case, to display that value in the MsgBox.

We can call constants almost anything we like. Choosing a meaningful name, like "Greeting" makes the code easy to read when we come back to it later.

Some words won’t work as constants. These are words which already have a meaning in VBScript. For example, you can’t call your constant "const". This will cause your script to error.


3.2 – Join two strings

Example

Change Main.vbs to:

Const Greeting = "Hi"

MsgBox Greeting & "!"

WScript.Sleep(2000)

MsgBox Greeting & " again!"

Run the script.

Explanation

The "&" symbol is known as ampersand or, informally, as the ‘and’ sign. It takes the values to either side of itself and joins them into a single text string, which MsgBox then displays.

The joining process is called "concatenation", so "&" is known as the concatenation operator. Similarly, the "=" is known as the assignment operator.

As shown here, constants can be used to avoid duplicating a value in a script. This makes the script easier to maintain. To change the greeting back to "Hello", for example, you now only need to change it in one place. Here, that only saves a little time, but what if the same value were being re-used in fifty places. The bigger a script gets, the more important ease of maintenance becomes.

Exercise 5

Change the greeting back to "Hello".


Answer 5

Const Greeting = "Hello"

MsgBox Greeting & "!"

WScript.Sleep(2000)

MsgBox Greeting & " again!"


3.3 – Use math

Example

Edit into Test.vbs. We’ll come back to Main.vbs later.

If you enter and run the following scripts:


	
MsgBox 2 + 2



	
MsgBox 1 - 2



	
MsgBox 10 * 0.2



	
MsgBox 10 / 3





You get the results:


	
4



	
-1



	
2



	
3.33333333333333





Explanation

To do calculations, use the arithmetic operators as above.

As shown, if you use division, you may get a very long, non-user-friendly, decimal number as a result.

Example

Now try this:


	
MsgBox 7 Mod 3





You get the result:


	
1





Explanation

The "Mod" operator gives the remainder of a division. In the example, we divide seven by three, and get a remainder of one. There is always a remainder. Even if the numbers are exact multiples, like "6 Mod 3", that just means the remainder is zero.

Example


	
MsgBox 1 + (2 * 3)



	
MsgBox (1 + 2) * 3





You get the result:


	
7



	
9





Explanation

If you want to make sure one part of a calculation happens first, use brackets (technically called parentheses). Operations inside brackets happen before operations outside of them.

In the examples above, the numbers and the operators are the same, but the brackets change the results.

In "1 + (2 * 3)" the "*" is inside the brackets, so we get two times three equals six, plus one equals seven.

In "(1 + 2) * 3" the "+" is inside the brackets, so we get one plus two equals three, times three equals nine.

Exercise 6

UK VAT (sales tax) is 20% (hint: that’s 0.2 as a decimal number). You are selling an e-book with a net price of £2. The customer will pay the gross price, i.e. net plus VAT.

Make Test.vbs calculate what the customer pays. To keep this exercise simple, just calculate the number, without currency units or extra zeroes after the decimal point. A script that displays "2.4" is fine. Show the result in a message.


Answer 6

You can use:

MsgBox 2 + (2 * 0.2)


3.4 – Use number constants

Example

Return to Main.vbs. Edit it to:

Const WaitTime = 2000

MsgBox "Click OK to wait for " & WaitTime & " ms."

WScript.Sleep(WaitTime)

MsgBox "Wait completed!"

Run it and see if it works

Explanation

Constants can hold numbers, instead of strings.

Notice that MsgBox makes the script wait until the user clicks OK. That’s why the two-second sleep doesn’t start until then.

Note for e-reader users

The scripts are starting to contain longer lines of code. Depending on the settings and screen-size of your e-reader, some lines of script may "wrap" around and appear as two lines on your screen. These lines still need to be entered as single lines in your script file. Fortunately, you can see when your e-reader wraps a line of script. It put a smaller space between the wrapped parts, the same as it does between lines of text in the same paragraph.

Exercise 7

Make the script display the wait time in seconds


Answer 7

With brackets, it’s easy to make this change:

Const WaitTime = 2000

MsgBox "Click OK to wait for " & (WaitTime / 1000) & " seconds."

WScript.Sleep(WaitTime)

MsgBox "Wait completed!"


3.5 – Allow user input

Example

Change Main.vbs to:

Dim msg

msg = InputBox("Enter a message:")

MsgBox "You said: " & msg

Run it and see what happens. Enter a message when prompted and then click OK.

Explanation

InputBox lets the script ask a question and receive an answer.

Dim is short for dimension, and is used to create, or "declare", a variable. In code, a variable holds a value, such as the text the user enters in the InputBox. Other parts of the code can then use that value.

This script declares the variable "msg". It prompts the user for a message using InputBox. The "=" sign, known as the assignment operator, takes the message returned by InputBox and assigns it to the variable.

The variable’s value is then re-used by MsgBox.

We use a variable, not a constant, to hold the message, because constants’ values can only be set by writing them in the script. That is, the value must be written into the code.

Variables’ values can be set by other parts of the code. This means they can be changed, e.g. by the user typing something different into InputBox.

Exercise 8

Make Main.vbs calculate the VAT (at 20%) on a net price, as entered by the user. For example, if they enter 10, display 2. As before, in your answer, ignore currency and fixed decimal places.


Answer 8

You could use something like this:

Const VatRate = 0.2

Dim net

net = InputBox("What is the net price?")

MsgBox "VAT: " & (net * VatRate)

At this point, you might wonder why I chose this example. How is a VAT calculation relevant to VBScript?

It can be. VBScript is not just a tool for Windows automation. VBScript is also the language of classic ASP, of the legacy ASP websites and intranet sites, which organisations still use to conduct business. Those have to be maintained, and that includes things like updating the code to cope with changes to VAT regulations.


3.6 – Avoid typos

Example

Edit Main.vbs, to:

num = 5

MsgBox num

Run it. It works. VBScript treats "num" as a variable. We can use variables without declaring with "dim" first.

Now misspell the second use of the variable:

num = 5

MsgBox nun

Re-run it.

This time, you’ll see a pop-up with no message. Why didn’t it error though? It didn’t error because VBScript just assumed "nun" was a variable too. However, as the nun variable doesn’t contain a value, MsgBox can’t display one either.

Now use dim:

dim num

num = 5

MsgBox nun

Re-run it.

The result will be the same. Using "dim" alone doesn’t solve this problem.

Now add one more line:

option explicit

dim num

num = 5

MsgBox nun

Re-run it.

You will see an error: "variable is undefined: nun".

Explanation

You don’t have to use "dim" to declare variables. If you don’t declare something, as with "nun", VB Script will guess that it’s a variable. Of course, as shown, you might end up with the wrong data, or no data at all.

In this example, the message makes the problem obvious. However, most scripts don’t have pop-ups. Most scripts do things like write to files or databases, or they run other programs. In those cases, you could make a typo, and not spot the problem for, potentially, ever. That is, until you notice that your script has been writing blank files for six months.

To guard against this, use Option Explicit. This enforces using "dim" to declare variables. If you then make a typing error (or forget to declare), it will tell you so. It will even tell you where in the file to fix it.

To use Option Explicit, just put it at the top of a script file.

This book doesn’t show Option Explicit at the start of every script. That would get very repetitive. You may wish to use it anyway though.


3.7 – Use a function

Example

Change the code in Test.vbs to:

MsgBox Now()

Run it and see what happens. You should see the current date and time appear.

Explanation

The "Now()" in the script above is an example of a function. A function is a program which returns a value to the script.

Exercise 9

There is a Time function. Write a script to show what it does.


Answer 9

MsgBox Time()


3.8 – Alter text

Example

Change Main.vbs to:

Dim msg

Dim shout

msg = InputBox("Enter message to shout")

MsgBox "Shout: " & msg & "?"

shout = UCase (msg)

MsgBox shout

Run it and see what it does.

Explanation

Focus on this line:

msg = UCase (msg)

The function here is UCase, short for Upper Case. It has a value, in the form of the msg variable, between its brackets. It takes this value as an input, called an "argument" or "parameter", and uses it to generate an output, or "return value". The return value is the argument string, but converted to upper case.

We assign the return value to the "shout" variable, which we show in the second MsgBox.

Exercise 10

Change "shout" to "whisper", and make the script show the message in lower case instead. Guess the name of the function you need. Hint: what could be short for "Lower Case"?


Answer 10

Dim msg

Dim whisper

msg = InputBox("Enter message to whisper")

MsgBox "Whisper: " & msg & "?"

whisper = LCase (msg)

MsgBox whisper


3.9 – Re-use a variable

Example

If you change the code in Main.vbs to this:

Dim msg

msg = InputBox("Enter message to whisper")

MsgBox "Whisper: " & msg & "?"

' We will now re-use the variable

msg = LCase (msg) ' variable re-used!

MsgBox msg

You can run it and see that it still works as before.

Explanation

Variables are just that: variable. They can hold different values at different times. In this case, we assign a message from InputBox and show it to the user. Then, we get the lower case version and make the variable store that instead.

Sometimes, when trying to find a problem in a script, you’ll want to know the value of a variable at a particular point. You can use MsgBox to make the script display the value at that point.

The phrases after the single-quotes (apostrophes) are called comments. We use the single-quote to tell the script to ignore the rest of the line. This lets us write our own notes, or "comments", into the file without causing an error.

Note

From now on, the scripts will be more varied and won’t always build on each other as closely as before, so they won’t be labelled as for Main.vbs or Test.vbs. You may keep editing these two files or save each script under a new name as you go. Not all the scripts are intended to be run, or edited as part of an exercise. Some are just provided as examples to be read, so don’t need to be typed out. Those will be flagged with phrases like "If you run:" and "Look at this script:".


4 – Control other programs


4.1 – Run another program

Example

Write the script below:

Dim Shell

Set Shell = Wscript.CreateObject("Wscript.Shell")

Shell.Run "chrome.exe"

Run it and see what happens.

Now, edit the script to:

Dim Shell

Set Shell = Wscript.CreateObject("Wscript.Shell")

Shell.Run "chrome.exe www.bing.com"

Be sure to type the space between "chrome.exe" and "www".

Run it and see what happens.

Explanation

The first script does the following:


	
Creates a shell object for interacting with Windows.



	
Makes the shell run Chrome.





The second script does the same thing, but also inputs the string "www.bing.com
" to Chrome. A string provided to a program this way is called a start-up parameter. Many programs accept start-up parameters. They use them however the program designers decided. In Chrome’s case, when it gets a parameter like this, it loads the webpage.

In this script, "Shell" is a variable. This variable holds the reference to the shell object we created using the "CreateObject" function.

We use the word "Set", along with the assignment operator "=", to make a variable hold a reference like this. This gives us a way to refer to the object later in the script. Then we can tell the object to do things, like run Chrome.

An object, put simply, is a set of programs (known as methods) and/or data (available through "properties") which we can re-use.

We used the "Run" method of the "Shell" object. Using a method is known as "calling" it.

Methods are classified as either functions or subroutines (subs). We’ll cover both types, and their differences, in more detail later.

Exercise 11

Make the script load a different web page, e.g.

www.amazon.com/D-Armstrong/e/B014GEOZ6A


Answer 11

Just change Chrome’s parameter:

Dim Shell

Set Shell = Wscript.CreateObject("Wscript.Shell")

Shell.Run "chrome.exe www.amazon.com/D-Armstrong/e/B014GEOZ6A"


4.2 – Interact with another program

Example

Write the script:

Dim Shell

Set Shell = WScript.CreateObject("WScript.Shell")

Shell.Run "notepad.exe"

WScript.Sleep 1000

Shell.AppActivate "Notepad"

WScript.Sleep 1000

Shell.SendKeys "Testing"

Shell.SendKeys "{ENTER}"

Shell.SendKeys Now()

Run the file. See what happens

Explanation

This script does the following:


	
Creates a shell object for interacting with Windows.



	
Makes the shell run Notepad.



	
Waits a second for Notepad to open.



	
Switches the active window to the Notepad application.



	
Waits a second for the switching to finish.



	
Sends the keystrokes to type "Testing", then the enter key (to start a new line), then the date, to the active window, i.e. Notepad.





SendKeys mimics keyboard input from the user. It allows scripts to interact with other programs, via their user interfaces. This can be used to automate work on legacy systems, or for testing purposes.

SendKeys can read a string and send each character as a keystroke, but some characters are not easily handled in strings. The "Enter" key is one such example. To send keys like enter, escape, caps lock, and so on, we enclose their names in curly braces, like so, "{ENTER}".

Exercise 12

Make the script load Chrome and have it search for "VBScript".

You may find, due to Chrome’s autocomplete feature, that you end up searching for something like "VBScripting" or "VBScript file". That’s close enough for this though.


Answer 12

This will do it:

Dim Shell

Set Shell = WScript.CreateObject("WScript.Shell")

Shell.Run "chrome.exe"

WScript.Sleep 1000

Shell.AppActivate "Chrome"

WScript.Sleep 1000

Shell.SendKeys "VBScript"

Shell.SendKeys "{ENTER}"


5 – Write to a file


5.1 – Write to a file

Example

Write the script:

Dim filePath, fso, file

filePath = "SomeText.txt"

Set fso = CreateObject("Scripting.FileSystemObject")

Set file = fso.CreateTextFile(filePath, True)

file.Write("File named: " & filePath & vbCrLf)

file.Write("Created at: " & Now())

file.Close

Run it, then open the file it creates and see what it says.

Explanation

This script creates a file and writes text to it.

It uses the CreateObject method to make a File System Object, which we reference using the variable "fso".

The file system object has the method we need to create a new file. We pass the method two arguments.


	
The address, or "path", to create the file at in the file system. As we only stated the file name, it ended up in the same folder as the script, by default.



	
A Boolean (true/false) value, to specify whether the file we are creating should overwrite any existing file, with the same name and path. We pass it True, to say, yes, it should. This lets us re-run the script without getting a "file already exists" error.





The method returns an object reference into the "file" variable, allowing the script to interact with the file object it just created.

We use the write method on "file" to write some text. We start a new line, at the end of the first write, by having the script write the new line character into the file. VBScript has a built-in constant for this character: vbCrLf (short for VB Carriage Return, Line Feed).

At the end, we close the file. We do this to make sure the file is unlocked on the system, so that other programs and users can work with it.

Also, notice how we declared multiple variables in one line, by separating them with commas.

Exercise 13

Change the script to create another text file, "MoreText.txt", with the word "Success!" inside.


Answer 13

Change the script to:

Dim filePath, fso, file

filePath = "MoreText.txt"

Set fso = CreateObject("Scripting.FileSystemObject")

Set file = fso.CreateTextFile(filePath, True)

file.Write("Success!")

file.Close

Save a copy of this script. We’ll re-use it later.


5.2 – Lay out method calls

Example

Below is another version of the example script for file-writing:

Dim filePath, fso, file

filePath = "SomeText.txt"

Set fso = CreateObject("Scripting.FileSystemObject")

Set file = fso.CreateTextFile(filePath, True)

With file

.Write("File named: " & filePath & vbCrLf)

.Write("Created at: " & Now())

.Close

End With

It works as before.

Explanation

If you have multiple calls to the same object, e.g. file, you don’t need to keep re-typing, or re-reading, the word "file" in your code:


	
file.Write



	
file.Close





The With...End With statement shown above lets you reference an object once, and use it repeatedly. Inside the With-block, you only need to type "." to reference the object.

The time saved is minimal here, but this is a tiny demo script. When you have ten or twenty lines referencing the same object, the With-block can be useful.

Exercise 14

Change the script from the file-writing exercise to use With...End With.


Answer 14

As below:

Dim filePath, fso, file

filePath = "MoreText.txt"

Set fso = CreateObject("Scripting.FileSystemObject")

Set file = fso.CreateTextFile(filePath, True)

With file

.Write("Success!")

.Close

End With


5.3 – Handle double-quotes in strings

Example

Write the script:

Dim filePath

filePath = "SomeText.txt"

MsgBox "File named: """ & filePath & """"

On the last line, that’s three quotes near the middle and four quotes near the end.

Run it. See what the message is.

Explanation

This is a message containing quotation marks.

Normally, quotation marks terminate strings in VBScript. To stop them ending the string and breaking our code, we have to type the "special" character twice. The code will then treat the doubled-up quotes as a single, normal character, i.e. one which doesn’t mark the end of the string.

Of course, we also need to terminate the string at some point.

That’s why the code uses three and four quotation marks in a row, to display quote characters around the file path.

Making a special character appear as itself in a string is known as "escaping" the character.

Exercise 15

Can you make MsgBox display a pair of double-quotes, together?


Answer 15

We need six quotes in a row for this:

MsgBox """"""


5.4 – Make a script write a script

Example

You don’t need an example for this. We have covered the necessary concepts already.

This exercise, and the one after it, are a bit more challenging, but have a go! Even if you end up skipping to the answers, they’ll still show what’s possible with what we already covered.

Exercise 16

Make a script create another script file, called "Greeting.vbs", which opens a message box and says "Hello!".

Test the greeting script to make sure it works.


Answer 16

I’d do it like this:

Dim filePath, fso, file

filePath = "Greeting.vbs"

Set fso = CreateObject("Scripting.FileSystemObject")

Set file = fso.CreateTextFile(filePath, True)

file.Write("MsgBox ""Hello!""")

file.Close


5.5 – Make a script run another script

Example

Again, no example needed here: you have the knowledge!

Exercise 17

Can you make your last script run the greeting script, after it creates it?


Answer 17

Just add these lines at the end (after the file-closing line):

Dim shell

set shell = WScript.CreateObject("WScript.Shell")

shell.run filePath


6 – Make decisions


6.1 – Compare numbers

Example

If you run:

MsgBox 1 = 1

MsgBox 1 <> 1

MsgBox 1 < 2

MsgBox 1 > 2

MsgBox 1 <= 2

MsgBox 1 >= 2

You get:

True

False

True

False

True

False

Explanation

The symbols used here are comparison operators. They take two values and compare them to return a True/False value, called a Boolean value.

The operators are:

= equals

<> not equal to

< less than

> greater than

<= less than or equal to

>= greater than or equal to

What an operator does is known as an operation. The values used in the operation are known as operands. In the example below, one and two are operands:

MsgBox 1 < 2

Exercise 18

Make a script which lets the user enter a number and checks whether it’s a multiple of 3. That is, return True if the number is a multiple, and False otherwise.


Answer 18

You could use something like this:

Const divisor = 3

Dim num, result

num = InputBox("Enter a multiple of " & divisor)

result = ((num Mod divisor) = 0)

MsgBox result

Note: we don’t actually need the brackets on the "result = …" line. They are just to make clear which order the math, comparison, and assignment operations happen in.


6.2 – Compare strings

Example

Write the script:

Const yourName = "D Armstrong"

Dim userName, result

userName = InputBox("Enter your name:")

result = (userName = yourName)

MsgBox "User is creator: " & result

Put your own name in the constant. Run the script and re-enter your name when prompted. Then re-run it and re-enter your name, with caps lock on.

This reports a match the first time but not the second.

Explanation

You can compare strings with comparison operators, using "=" to check if they match. Equally, you can use "<>" to check if they don’t match. However, the comparison is case-sensitive.

Exercise 19

Make the script match despite different cases in the strings.


Answer 19

One way is to convert both strings to upper case before comparing:

Const yourName = "D ARMSTRONG"

Dim userName, result

userName = InputBox("Enter your name:")

userName = UCase(userName)

result = (userName = yourName)

MsgBox "User is creator: " & result


6.3 – Use logic

Example

Write the script:

Dim age, isValid

age = InputBox("Enter your age")

isValid = (age >= 5 and age <= 120)

MsgBox "Age valid: " & isValid

Run it with your age, then some really old and really young ages. See what it says.

Explanation

This script only believes you if you say your age is between 5 and 120.

It combines the Boolean values returned by two comparisons using the "and" operator. The operator returns true if both values are true, and false otherwise.

There is also an "or" operator. This returns true if either of its operands are true. It only returns false if neither is.

Exercise 20

Make the script check if the age you provide is invalid. Use the "or" operator.


Answer 20

You could do this:

Dim age, isInvalid

age = InputBox("Enter your age")

isInvalid = (age < 5 or age > 120)

MsgBox "Age invalid: " & isInvalid


6.4 – More logic

Example

If you run:

MsgBox Not (1 = 2)

The message is "True".

Explanation

The script compares 1 and 2 to return FALSE, then passes that to the NOT operator, which returns the opposite of whatever Boolean value you give it: true for false, false for true. This operator only takes one value.


6.5 – Decide what to do

Example

Write the script:

Dim num, result

num = InputBox("Enter a number:")

If num > 0 Then

result = num & " is a positive number."

Else

result = num & " is a negative number."

End if

MsgBox result

Run the script and enter a positive number when prompted. Repeat with a negative number. Repeat with zero.

Explanation

This script checks whether the number you enter is above zero or not, and reports back.

The If-statement, as shown above, tells a script to make a decision based on a condition we define, e.g. "num > 0". The condition can be anything which returns true or false.

If the condition is met, i.e. true, the script runs the code in the Then-clause. If not, it runs the code in the Else-clause.

It lets a script behave differently in different situations.

The Else-clause is optional, so if the script only needs to do something different when the condition is met, the Else-clause can be left out.

Code which is part of a conditional clause, like "then" or "else", is usually indented to show this. When scripts get larger, they may have multiple If-statements, some inside of others. Indents make it easier to see which lines are part of the same clauses. They make the code more readable.

Exercise 21

Notice how the script wrongly classifies zero as negative. It does this because zero is not " > 0". We can fix this with another If-statement, so we get "0 is zero". Can you work out how?


Answer 21

You could do this:

Dim num, result

num = InputBox("Enter a number:")

If num > 0 Then

result = num & " is a positive number."

Else

If num = 0 Then

result = num & " is zero."

Else

result = num & " is a negative number."

End if

End if

MsgBox result

The second If-statement runs as the Else-clause of the first. An If-statement inside another is said to be "nested".


6.6 – Do something repeatedly

Example

Write the script:

Dim i

For i = 1 To 3

MsgBox "message"

Next

Note: the "i" is a variable, so if using "option explicit", you must declare it with "dim" before the loop.

Run it and see what happens.

Now change the script to:

Dim i

For i = 1 To 3

MsgBox "message " & i

Next

Run the new script.

Explanation

This is called a for...next loop. It counts from one number to another and with each step, runs the code inside itself. The count is stored in a variable, in this case "i", so it can be used by the code inside the loop, as the second script showed.

Exercise 22

The game fizz-buzz is often used as a simple programming test.

The fizz part of the game is as follows:


	
start counting from one, saying the numbers



	
if a number is a multiple of three, say "fizz" instead





Can you change the script to do this? Make it count as high as seven.


Answer 22

Put an If-statement inside your loop:

Dim i

For i = 1 To 7

If i Mod 3 = 0 Then

MsgBox "fizz"

Else

MsgBox i

End If

Next


6.7 – Decide between more options

Example

Edit the last script to:

Dim i

For i = 1 To 7

If i Mod 3 = 0 Then

MsgBox "fizz"

ElseIf i Mod 5 = 0 Then

MsgBox "buzz"

Else

MsgBox i

End If

Next

Run it and see what it does.

Explanation

This is a fizzbuzz script!

The ElseIf clause lets you check extra conditions in an If-statement. You could achieve the same thing with a nested If-statement, but this is shorter and tidier.

You can have as many ElseIf clauses as you like. They can each check any condition. However, those conditions will be checked from top to bottom, and only the first condition met will have its Then-clause run.

Exercise 23

The script isn’t perfect. If you allow it to count to fifteen, it will say "fizz", because fifteen is a multiple of three. However, fifteen is also a multiple of five. In this case, the script should say "fizz buzz".

Make it do so.

To save time on testing this, you may want to start the count from ten, as in:

For i = 10 To 15

This range lets you test that “fizz”, “buzz”, “fizz buzz” and normal numbers all get displayed, without sitting through fifteen pop-up messages.


Answer 23

One way to do it:

Dim i

For i = 10 To 15

If i Mod 3 = 0 and i Mod 5 = 0 Then

MsgBox "fizz buzz"

ElseIf i Mod 3 = 0 Then

MsgBox "fizz"

ElseIf i Mod 5 = 0 Then

MsgBox "buzz"

Else

MsgBox i

End If

Next


6.8 – Make decisions with less code

Example

Look at this script:

dim day

day = Weekday(Now())

Select Case day

Case 1

MsgBox "Sunday"

Case 2,3,4,5,6

MsgBox "Weekday"

Case 7

MsgBox "Saturday"

Case Else

MsgBox "Not a real day"

End Select

It returns a message depending on what day of the week it is.

Explanation

The Weekday function returns a day number for a given date. It counts days from Sunday, so Sunday is 1, Monday is 2, etc.

The Select Case statement takes a variable, in this case "day". It checks each "case" statement, one after the other, for a match. If it finds one, it run the code for that case.

The Select Case statement is based on finding an exact match to a single variable. This makes it less flexible than Else-If. However, it can be a lot more concise and easier to read. If that’s not apparent, try, or just imagine, re-writing the example above as an ElseIf statement. You’d have to keep duplicating the variable name and comparison operators.


6.9 – Generate a big file

Example

No example needed, we have covered enough VBScript to do this.

Exercise 24

Suppose you want to generate a file containing lots of text. You need it to test out a file-reading script, which you plan to create later. How would you do it?

You could create a file and just keep writing the same line to it. You could even number those lines.

Try that.

Make your script generate a file with a thousand lines, like so:

1 – dummy text

2 – dummy text

…


Answer 24

You could do this:

Dim filePath, fso, file, i

filePath = "BigFile.txt"

Set fso = CreateObject("Scripting.FileSystemObject")

Set file = fso.CreateTextFile(filePath, True)

For i = 1 to 1000

file.Write(i & " - dummy text" & vbCrLf)

Next

file.Close

Of course, a thousand of these lines still isn’t a very big file, but we can always increase the number.

Exercise 25

Change the script to generate "BigFile.html". Make every tenth line a html comment too, that is, it should begin with "<!--" and end with "-->". Make other lines into html paragraphs, that is, they start with "<p>" and end with "</p>".

Open the resulting Html file. Windows recognises the ".html" extension and automatically opens the file in your browser, e.g. Edge, Chrome, etc. You should see all lines displayed, except for the comments.


Answer 25

You could do this:

Dim filePath, fso, file, i

filePath = "BigFile.html"

Set fso = CreateObject("Scripting.FileSystemObject")

Set file = fso.CreateTextFile(filePath, True)

For i = 1 to 1000

If i Mod 10 = 0 Then

file.Write("<!--" & i & " - dummy text" & "-->" & vbCrLf)

Else

file.Write("<p>" & i & " - dummy text" & "</p>" & vbCrLf)

End If

Next

file.Close


7 – Work with loops


7.1 – Repeat an action until something happens

Example

Write the script:

Dim num : num = 1

Do While num < 10

num = num * 2

MsgBox num

Loop

Run it.

Explanation

This script keeps doubling, and displaying, a number, until it exceeds ten.

This kind of loop is known as a Do...Loop. It is often used with a While-condition, as shown here (and so referred to as a Do...While loop). The loop checks the condition, and if that condition is met, runs the code inside, then goes back to the "Do" line and repeats the cycle, until the condition returns false.

Beware though, if you give it a condition which never returns false, the script will just keep going round the loop.

Notice also how we declare a variable and assign it a value on the same line, by using the ":" (colon) character.

Exercise 26

Make the code only display the number if it is ten more.


Answer 26

This would work:

Dim num : num = 1

Do While num < 10

num = num * 2

If num >= 10 Then

MsgBox num

End If

Loop

But this would be better:

Dim num : num = 1

Do While num < 10

num = num * 2

Loop

MsgBox num

The first version duplicates the comparison, so that it happens twice for each cycle of the loop, which is inefficient.

This loop only repeats a few times, so it’s not noticeable. However, as loops run repeatedly, it’s generally best to minimize the code inside. That way, scripts have less work to do, so they finish faster.


7.2 – Understand While...Wend loops

Example

The loop below doubles a number until it exceeds ten:

Do While num < 10

num = num * 2

Loop

You could do the same with this:

While num < 10

num = num * 2

Wend

Explanation

The While...Wend loop is an older form of loop. VBScript continues to support it, for backwards compatibility (i.e. with old scripts that people wrote when While...Wend was
 in use). In examples like the above, it works the same way as Do...While.

However, you can do more with Do...While:


	
You can stop part way through a loop cycle, using "Exit Do". We’ll cover that later.



	
By moving the "While" and its condition to after "Loop", you can check the condition at the end of each cycle instead of the start. That way, the loop always runs through at least once.





You don’t need to use While...Wend, but it’s good to know what it does. You may find yourself working with older scripts at some point.


7.3 – Stop a script caught in an infinite loop

Example

This script has an infinite loop (no need to run it):

Dim x

x = 1

Do While x < 3

MsgBox x

Loop

Explanation

If you ran the script, it would keep giving the message "1", because there is no code to increase the value of the variable "x".

If you find that a script won’t stop, you can stop it yourself as follows:


	
Right-click on the task bar (normally at the bottom of your screen).



	
Select Task Manager.



	
In task manager, make sure the processes tab is open (it should be already, it’s the default).



	
In this tab, you will see a list of apps, followed by a list of background processes and a list of Windows processes.



	
On the apps list, find "Microsoft Windows Based Script Host". This is the program which runs VBScripts.



	
Right-click on it’s name. A context menu will appear.



	
On the menu, select End Task.






7.4 – Understand Do...Until Loops

Example

The loop below doubles a number until it exceeds ten:

Do While num < 10

num = num * 2

Loop

You could also do this:

Do Until num >= 10

num = num * 2

Loop

Explanation

The second loop checks the opposite condition to the first loop. However, because it uses "Until" instead of "While", it doesn’t stop until the condition is met (i.e. true). This lets us get the same end result.

This is known as a Do...Until loop. It’s like a Do...While, but it stops when the condition is met, whereas Do...While stops when the condition isn’t met.

It’s effectively the same as:

Do While Not...

Apart from this, it works the same way as Do...While:


	
You can use "Exit Do".



	
You can move "Until" and the condition to the end of the loop.






7.5 – Get part of a string

Example

If you run:

MsgBox Mid("Learn VBScript Fast", 2)

You get:

"earn VBScript Fast"

If you run:

MsgBox Mid("Learn VBScript Fast", 1, 14)

You get:

"Learn VBScript"

If you run:

MsgBox Mid("Learn VBScript Fast", 1, 1)

You get:

"L"

Explanation

The "Mid" function gets you text from midway through a string. It takes arguments as follows:

Mid(string, start [, length])

The "start" argument says which character position to start from, e.g. "L" is at position one, "e" is at two, etc.

The "length" argument is optional. That’s why it’s shown here with square brackets around it. You don’t type the square brackets though. They’re just a common way of marking something as optional in programming language documentation. If you look up any VBScript functions online, you’ll see the same thing.

If you use the "length" argument, the function returns that number of characters. Really, it returns up to
 that number of characters, as it may reach the end of the string first. If you don’t use the argument, the function will just return the rest of string.

Exercise 27

Make the function return "VB".


Answer 27

Like this:

MsgBox Mid("Learn VBScript Fast", 7, 2)


7.6 – Do more with strings

Example

Write the script:

Dim input

input = InputBox("Enter some text")

MsgBox Len(input)

Run it. Try entering a few different input strings when prompted. See if you can work out what the function does.

Explanation

The function "Len()" is short for "Length". It counts the number of characters in the string, including any spaces.

Exercise 28

Change the script to get the last character of a string. Use Len and Mid.


Answer 28

Like this:

Dim input, startPosition

input = InputBox("Enter some text")

startPosition = Len(input)

MsgBox Mid(input, startPosition)

The length of a string is also the position of its last character

Exercise 29

Change the script to get the last three characters of a string.


Answer 29

Hopefully, you figured out something like this:

Dim input, startPosition

input = InputBox("Enter some text")

startPosition = Len(input) - 2

MsgBox Mid(input, startPosition, 3)

By combining string functions, we can work with strings in all kinds of ways.

In this case though, there’s an existing function we can use instead. In the script, after InputBox, try using this:

MsgBox Right(input, 3)


7.6 – Leave a loop early

Example

Look at this script:

Dim input, position, length, char, result

result = "phone number"

input = InputBox("Enter telephone number or email:")

position = 1

length = Len(input)

Do Until position > length

char = Mid(input, position, 1)

If char = "@" Then

result = "email"

Exit Do

End If

position = position + 1

Loop

MsgBox "Thank you for entering your " & result

Explanation

This script decides whether a string is a phone number or email, based on whether it has an "@" character in. It’s not a good way to validate phone numbers, or emails, and we could just use the InStr function (short for “In String”) to find the "@" instead. We’ll cover that later. However, this way does show Exit Do in action.

The script starts by setting the result to "phone number", so that the result will default to this.

It takes an input string and checks it length (number of characters) using the "Len()" function.

It then loops through the string, one character at a time, checking if any of them are "@". If one is, it changes the result variable to "email" and leaves the loop early. The "Exit Do" makes it jump to the line after "Loop". If it gets to the end of the loop without finding a "@", it leaves with the default result still set.

Either way, it displays a result at the end.

Note: you can leave For-loops early in a similar way. Just use "Exit For".


8 – Work with text in files


8.1 – Search text

Example

Try out this script:

Dim searchTerm

searchTerm = InputBox("Enter search term:")

Msgbox InStr("I love baked beans", searchTerm)

Run it, enter "I" when prompted (make sure it’s a capital "I"), and see what happens.

Repeat with "love", then "macaroni".

You should see the results: 1, then 3, and then 0.

Explanation

The InStr function takes two arguments as strings and searches the first for the second. If it finds a match, it returns the position of the first matching character:

"I" is the first character in "I love baked beans", so you get 1.

"love" starts at the third character in "I love baked beans", so you get 3.

If there’s no match, such as for "macaroni", the function returns zero.

You can use the function to find the position of one string in another, or just to check if one string contains the other.

Exercise 30

Make the script display "True" if the phrase contains the search term and "False" if not.


Answer 30

Just add a comparison:

Dim searchTerm

searchTerm = InputBox("Enter search term:")

Msgbox InStr("I love baked beans", searchTerm) > 0


8.2 – Read a file

Example

Write the script:

Dim num : num = 0

Dim fso : Set fso = CreateObject("Scripting.FileSystemObject")

Dim f : Set f = fso.OpenTextFile("BigFile.txt")

Dim line

Do Until f.AtEndOfStream

line = f.ReadLine

If inStr(line, "00 ") Then

num = num + 1

End If

Loop

f.Close

MsgBox num & " lines contain matching text"

Note the space between the 00 and the closing quotation mark.

Run it and see what it does.

Explanation

This script reads a file, line-by-line, and counts the lines containing the digits "00" followed by a space: line 100, line 200, and so on.

The script uses the OpenTextFile method to return a TextStream object, which we reference via the variable "f". The text stream object reads through the file and streams it to the script.

If this sounds a bit technical, think of streaming videos online. You don’t have to download a whole video to start streaming; the computer plays the video from the start and downloads more as you go along. TextStream works like that, only with reading instead of downloading.

The loop checks the AtEndOfStream property, which returns true or false. If the TextStream object is at the end of the stream, i.e. the end of the file, it returns true. It could return true right away, if the file were empty. The file isn’t empty though, so the loop does run.

In the loop, the ReadLine method does two things:


	
It returns the first line of text, which is then assigned to the variable.



	
It gets ready to read the next line in the stream, if called again.





The script then checks the "line" variable and increments (or not) the count in "num".

The script keeps going round the loop, until the AtEndOfStream property returns true. Then there is no more file to read, so we make the script stop looping.

Exercise 31

Make the script read the file until it finds a line containing "99 ". If it finds one, display that to the user as a message, then stop.

Once you have got this working, create an empty text file, and run the script on that, instead of "BigFile.txt". Make sure it runs without errors or pop-up messages.


Answer 31

As below:

Dim fso : Set fso = CreateObject("Scripting.FileSystemObject")

Dim f : Set f = fso.OpenTextFile("BigFile.txt")

Dim line

Do Until f.AtEndOfStream

line = f.ReadLine

If InStr(line, "99 ") Then

MsgBox line

Exit Do

End If

Loop

f.Close


8.3 – Edit a file

Example

Edit your script file, to:

Dim fso : Set fso = CreateObject("Scripting.FileSystemObject")

Dim f : Set f = fso.OpenTextFile("BigFile.html")

Dim f2 : Set f2 = fso.CreateTextFile("BigFile2.html", True)

Dim line

Do Until f.AtEndOfStream

line = f.ReadLine

line = UCase(line)

f2.Write(line & vbCrLf)

Loop

f.Close

f2.Close

Run it and see what it does.

Explanation

This script copies lines from one file to another, but in upper case. It shows the basics of how to make an altered version of a file.

Exercise 32

Suppose we want to optimize a web page for faster download. We could remove all the comments to reduce the amount of data the page contains.

Make a script to create a copy of the Big Html file, with the comments removed.


Answer 32

One way to do this is:

Dim fso : Set fso = CreateObject("Scripting.FileSystemObject")

Dim f : Set f = fso.OpenTextFile("BigFile.html")

Dim f2 : Set f2 = fso.CreateTextFile("BigFile2.html", True)

Dim line

Do Until f.AtEndOfStream

line = f.ReadLine

If Not (Mid(line, 1, 4) = "<!--") Then

f2.Write(line & vbCrLf)

End If

Loop

f.Close

f2.Close


9 – Use arrays


9.1 – Use an array

Example

Write the script:

Dim a

a = Array("Mon","Tue","Wed","Thu","Fri")

MsgBox a(1)

MsgBox a(2)

Run it and see what happens.

Explanation

The script displays "Tue" then "Wed".

An array is a variable which holds a collection of data. This array, "a", has five "items" (or "elements") because we gave five arguments to the "Array" function which created it. As with variables, we can get, or set, the values the items hold.

You can refer to an array item using its position in the array. This position is called the "index". The index is 0-based, so to get "Mon", you would use:

MsgBox a(0)

Exercise 33

Change the last line of the script, to display the last item in the array.


Answer 33

Use:

MsgBox a(4)

Don’t use:

MsgBox a(5)

The second version causes a "subscript out of range" error. This means you tried to access a position which doesn’t exist.

Exercise 34

See if you can guess how to change "Tue" to "Tuesday" in the array (after the array function has created it though, not by using the array function). Display it with MsgBox, so you can see if it worked.


Answer 34

You can assign to an array item, the same as any other variable:

Dim a

a = Array("Mon","Tue","Wed","Thu","Fri")

a(1) = "Tuesday"

MsgBox a(1)


9.2 – Loop through an array

Example

Edit the script, to:

Dim a

a = Array("Mon","Tue","Wed","Thu","Fri")

MsgBox UBound(a)

Run it and see what happens.

Now add "Sat" and "Sun" arguments into the Array function. Run the script again.

Explanation

The UBound function returns "4" the first time, then "6". It returns the highest index number in the array. UBound is short for Upper Bound.

Exercise 35

You know how to get the last index. You know how to reference array elements. You know how For...Next loops work.

Can you put it all together and work out how to loop through the array items?

Join them all together into one string, with spaces between them, and display the result in a message box.


Answer 35

This would be good enough to display in a pop-up:

Dim a, i

a = Array("Mon","Tue","Wed","Thu","Fri","Sat","Sun")

Dim lastIndex

lastIndex = Ubound(a)

Dim msg : msg = ""

For i = 0 To lastIndex

msg = msg & " " & a(i)

Next

MsgBox msg

There’s a minor flaw: the script puts an extra space at the start of the message. That’s easy to deal with. Just put this line after the loop:

msg = Trim(msg)

The trim function removes extra spaces from the start and end of a string.

Also, you may be thinking that we could simplify this script. We could just make the loop "For 0 To 6", and avoid using UBound.

In this simple demo script we could, but UBound is worth knowing about. The benefit of using UBound is that it checks the array size for us, every time. Often, arrays are generated from other sources, such as:


	
splitting a string



	
querying a database



	
listing files in a folder





In these cases, we won’t know what size the array will be. It may even change when you run the script at different times, so we can’t just write a number into the script. UBound solves this problem.


9.3 – Loop through an array more easily

Example

Edit your script, to:

Dim week, day

week = Array("Mon","Tue","Wed","Thu","Fri","Sat","Sun")

Dim msg : msg = ""

For Each day In week

msg = msg & " " & day

Next

MsgBox msg

Run it and see what happens.

Explanation

This does the same thing as the last script.

The For Each...Next loop lets you loop through each item in a collection, such as this array.

Each cycle, the loop takes the next item from the array, and stores it in the variable we defined, in this case, "day".

Looping through an array allows us to process its items however we choose. If we do just want to join the items together though, there is an existing function which can do that for us:

Dim msg : msg = Join(week)

The Join function returns the array items, separated by spaces, as a string.


9.4 – Do more with loops

Example

You might wonder why you would ever want to loop through an array using For...Next, when For Each...Next exists. What can For...Next do that For Each...Next can’t?

Look at this:

Dim num

For num = 1 To 10 Step 2

MsgBox(num)

Next

Explanation

This displays all the odd numbers in its range. If you used "num" as an array index, you could process all the array items with odd positions.

The "Step" keyword lets us make a loop which goes up in twos, or threes, or more. The step value gives you more control than you can have with a For Each...Next loop. There’s more though.

Example

See below:

Dim num

For num = 10 To 1 Step -1

MsgBox num

Next

Explanation

This displays all the numbers in its range, but in reverse order. If you used "num" as an array index, you could process all the array items from last to first.

Exercise 36

Display all the even numbers, from highest to lowest, in the 5 to -5 range.


Answer 36

Five’s not even, so start at four:

Dim num

For num = 4 To -4 Step -2

MsgBox num

Next


9.5 – Do more with arrays

Example

You can also declare arrays as follows:

Dim a(3)

Dim num

For num = 0 To 3

a(num) = num * 12

Next

MsgBox (Join(a))

If you run this, you get the twelve times table, from 0 to 36.

Explanation

When we declare an array, we can set the upper bound at the same time. As shown though, the size of the array will be one item more than that.

Declaring an array this way though, means that its size is fixed and can’t be changed. If we use the array function instead, we get a dynamic array, which we can re-size. We’ll cover that later.

Example

If you put the following variable declaration in a new script:

Dim a(4,9)

If you run the script, it doesn’t error. What can this mean?

Explanation

It’s possible to have a 2-dimensional array (sometimes called a matrix). This example gives you a five by ten array. Mostly though, we only need 1-dimension.


9.6 – Do even more with arrays

Example

If you run this:

Dim a : a = Array(0,1,2,3)

a(4) = 4

MsgBox a(4)

You get a "subscript out of range" error.

If you run this:

Dim a : a = Array(0,1,2,3)

ReDim a(4)

a(4) = 4

MsgBox a(4)

This works.

Explanation

The "ReDim" statement re-declares an array, meaning you can change the size. It has a side-effect though...

Example

If you run:

Dim a : a = Array(0,1,2,3)

ReDim a(4)

a(4) = 4

MsgBox Join(a)

You only see "4" in the message.

Explanation

Re-declaring an array wipes out its existing values.

Example

If you run:

Dim a : a = Array(0,1,2,3)

ReDim Preserve a(4)

a(4) = 4

MsgBox Join(a)

You get all the values in your message.

Explanation

The "Preserve" keyword protects existing values when you re-declare an array.


10 – Use and create functions


10.1 – Give the user a choice

Example

Write the script:

Dim choice : choice = MsgBox("Demo ""Replace()"" Function?", vbYesNo)

If choice = vbYes Then

Dim input : input = InputBox("Enter some words:")

MsgBox Replace(input, " ", "_")

Else

MsgBox "OK. Bye!"

End If

In the "Replace()" function above, the second argument used is a space, the third is an underscore character.

Run it and try it out.

Explanation

The script displays a message with buttons for the user to click yes or no. It runs the demo if they click yes.

The MsgBox we have been using in this book is a function. It has an optional argument, for which we are now entering the built-in constant "vbYesNo". Doing this makes the message box give the user a choice of buttons (Yes and No) to click on.

If the user clicks Yes, the MsgBox function returns the built-in constant vbYes. If not, it returns the vbNo constant. The script uses the return value to decide whether to show the demo or not.

Notice that we use brackets around the MsgBox arguments here, where we didn’t before. Before, we weren’t using the return value of MsgBox, so brackets were optional. Now that we are using the return value (by assigning it to the "choice" variable) brackets are required.

If the demo goes ahead and you enter some text, the replace function searches that text for spaces and replaces them with underscores.

Note: The replace function isn’t limited to single characters. If you wanted to replace the word "red" with "yellow", you could do this:

Replace(input, "red", "yellow")

Exercise 37

You have just found out that there is an alternative to vbYesNo called vbOkCancel.

Make the script work with that.


Answer 37

Just change the constants:

Dim choice : choice = MsgBox("Demo ""Replace()"" Function?", vbOkCancel)

If choice = vbOk Then

Dim input : input = InputBox("Enter some words:")

MsgBox Replace(input, " ", "_")

Else

MsgBox "OK. Bye!"

End If


10.2 – Create your own function

Example

Write the script:

Function DeSpace ( input )

Dim newChar : newChar = "_"

DeSpace = Replace(input, " ", newChar)

End Function

MsgBox( DeSpace ( "I am a filename.txt" ) )

Run it and see if it works.

Explanation

You just created and used your own function!

This "DeSpace" function takes one parameter, which we called "input", though we could call it anything. It passes that on, into the replace function, along with the two arguments we set inside of our function. It assigns the result of that into the "DeSpace" variable.

Notice how the variable has the same name as the function, and isn’t declared with "Dim". When we create a function, we automatically declare a return variable with the same name.

When the code reaches "End Function", the function returns whatever value is in its return variable at the time.

There is also an "Exit Function" (not shown here). This lets you leave the function part-way through, much like how Exit Do works with loops.

The benefit of a function is that you can re-use the process it contains, elsewhere in your script, as many times as you like, without duplicating the code.

Also, notice that the function can be in the same script file, but won’t run unless and until another part of the script calls it.

And yes, in our function, we could have put the "_" string into the replace function directly, and saved a line of code, but I wanted to show that functions could do multiple things, on multiple lines, not just return a value. If you want, you can use If-statements, loops, even objects, inside a function.

Example

On the end of the script, add the line:

MsgBox newChar

Run it.

You won’t see the character appear in the second message box. Instead, you’ll get an error message: "variable is undefined". Variables declared inside a function only exist within it. They are limited to the function’s "scope".

Exercise 38

The following code displays
 the name of the current day:

MsgBox WeekDayName(WeekDay(Now()))

Make a function to return
 the same value. Call the function "CurDay". Don’t define any arguments. Once you’ve made the function, display it’s result using MsgBox.


Answer 38

This works:

Function CurDay()

CurDay = WeekDayName(WeekDay(Now()))

End Function

MsgBox CurDay()

So does this:

Function CurDay

CurDay = WeekDayName(WeekDay(Now))

End Function

MsgBox CurDay

Notice how we don’t need brackets in the function definition, or in the function call. We only need brackets when enclosing arguments. If there are no arguments, brackets are optional (and redundant).

We only have to use brackets when calling a function if:


	
it has arguments, and



	
we assign (i.e. use) the return value.






10.3 – Make a more complex function

Note

This chapter is more complex than the rest of the book, and shows a general programming technique (recursion), not an essential part of VBScript. Also, its exercise is probably the trickiest one in the book. Please consider this chapter as an optional extra, and if it’s not of interest, just skip over it.

Example

The example has some underlined text. Ignore that extra formatting for now, and just type it as normal.

Write the script:

Function Compress( input )

' second argument in Replace() is a double space, third is a single space

Compress = Replace(input, "  ", " ")

If Compress <> input Then

Compress = Compress
(Compress)

End If

End Function

MsgBox Compress(
"1 2  3   4    end")

For the last line, after typing each number, type the same number of spaces.

Run the script. It will display the message, but without the extra spaces. No matter how many spaces were in each group, they are all now single spaces.

Explanation

The function replaces double-spaces in a string (with singles), then replaces any double-spaces in the result (e.g. those compressed from quadruple-spaces), and keeps going until all spaces are single. It returns the "compressed" result.

This code looks complex, but it isn’t. The fact that VB Script re-uses function names as variables does make it harder to read though, so I’ve underlined the function calls, to make them stand out from the variable name usage.

The function does the following:


	
replace any double-spaces with single-spaces



	
check if the result is different



	
if so, run the same process on the result



	
if not, continue to "End Function" and return the result.





If this sounds a lot like a loop, it is! It’s just that the loop isn’t in the function.

The function calls a copy of itself which calls a copy of itself, and so on, replacing double-spaces each time. When there are none left, the replace function doesn’t change the string any more, meaning the input and result match. At that point, whichever copy of the function is running ends. In doing so, it passes the result back to the function which called it, which then ends and passes the result on, and so on.

When the original copy (or "instance") of the function ends, it passes the result back to MsgBox.

This looping technique is called recursion, and its commonly supported across many programming languages. It works by opening an instance of a function for each cycle of a loop. This does have a limit though. The script can only open so many copies at a time. After that, it gets a "stack overflow" error.

Recursion works here, because file names tend to be relatively short. In general though, for VBScript, its better to practise using For and Do loops. Those loops work inside a single function and won’t cause a stack overflow error.

Exercise 39

Make a function that takes a string and returns it with any groups of spaces turned into single-spaces. Use a Do...Loop instead of recursion. You may find it easier to work from the bullet points than to convert the code above directly.


Answer 39

One way is:

Function Compress( input )

Do While True

Compress = Replace(input, "  ", " ")

If Compress = input Then

Exit Do

End If

input = Compress

Loop

End Function

MsgBox( Compress(
"1 2  3   4    end") )

By using the value True directly in the loop’s While-condition, we tell the loop to just keep going (until the If-statement runs the "Exit Do" line).

However, the While-condition is actually an optional part of the Do...Loop statement, so we can just leave out "While True" and get the same result.

Other than that, it’s the same procedure:


	
Process the input to get a result (stored in Compress)



	
Compare the result to the input, to see if any double-spaces were replaced



	
If so, copy the result into the input variable and repeat the process



	
If not, stop the loop and return the result






10.4 – Preserve your inputs

Example

Shown below is a shorter, less-thorough version of the Compress function. It also compresses the letter "x", not spaces, but only because that’s easier to see on screen. You’ll need to be able to count the characters, for this example to be useful.

Edit the script, to:

Function Compress( input )

Compress = Replace(input, "xx", "x")

End Function

In the same file, add:

Function DeX ( text )

text = Compress ( text )

DeX = Replace(text, "x", "_")

End Function

This is just the "DeSpace" function, except modified to:


	
replace "x"s instead of spaces



	
compress double "x"s first





Finally, add to the script:

Dim startValue : startValue = "2xx2xxend"

Dim result : result = DeX(
 startValue )

MsgBox( startValue )

MsgBox( result )

Run the function and notice how many "x"s there are in first message box.

Wait a minute! We set the start value to have double "x"s. Why does MsgBox display singles?

Explanation

The start value has been changed.

The only thing that could have done it is the "DeX" function, but how?

The script passed the "startValue" variable to the function as the "text" argument. The function compressed the text argument’s string, and assigned the result back into the "text" argument (arguments being variables, it can do this).

However, when it updated that argument, it also updated the "startValue" variable.

The function doesn’t copy the value of "startValue". It simply makes "text" a reference to the same data which "startValue" holds. That is, it points to the same location in computer memory. The variables are like two doors to the same room. What you put into one is what’s available to take out of the other.

This way of copying variables into functions is called passing by reference. If we want to re-use the variables in the function calls, it’s a problem.

The solution is simple though, you can:

avoid assigning to argument variables (inside functions you write), or

pass arguments by value instead

To make a function take an argument by value, we add the keyword "ByVal" before that argument in the function definition:

Function DeX ( ByVal text )

This way, the value is copied, not the reference, and any changes to the "text" argument, won’t affect the original variable ("startValue").

If we did want to affect the original variable, we could do this instead:

Function DeX ( ByRef text )

This makes it obvious, when reading the function’s code, that passing by reference is intended.

Example

Make the DeX() function take arguments by reference again.

Change the line which calls the DeX function, to this:

Dim result : result = DeX(
 (startValue) )

Run the script again.

You will see in the first MsgBox that, this time, the start value does not change.

Explanation

You can pass arguments in by value even if they are defined as "ByRef". Just wrap the argument in an extra pair of brackets.


11 – Work with the file system


11.1 – Rename a file

Example

Write the script:

Dim fso

Set fso = WScript.CreateObject("Scripting.FileSystemObject")

fso.MoveFile "I am a file.txt", "I_am_a_file.txt"

In the same folder as the script, create the file "I am a file.txt".

Run the script.

Check the file has been renamed to "I_am_a_file.txt".

Explanation

Often, spaces in file names require some kind of special handling by other programs, like Excel, or the Windows Cmd Line. It’s a bit like how quotation marks in strings have to be escaped in VB Script.

One way to avoid this is to not put spaces in file names. Some people use underscores instead.

Example

Rename the text file to what it was before.

In the same folder as your script file, create a subfolder called "MySubFolder".

Change the script to:

Dim fso

Set fso = WScript.CreateObject("Scripting.FileSystemObject")

fso.MoveFile "I am a file.txt", "MySubFolder\I_am_a_file.txt"

Re-run the script.

The file will not only be renamed, but moved into the subfolder.

Explanation

The "MoveFile" method changes the whole file path, which means you can use it to change which folder the file appears in.


11.2 – Create a subroutine

Example

Edit you script, to:

Sub RenameMyFile

Dim fso

Set fso = WScript.CreateObject("Scripting.FileSystemObject")

fso.MoveFile "I am a file2.txt", "I_am_a_file2.txt"

End Sub

RenameMyFile

In the same folder as your script file, create the file "I am a file2.txt".

Run the script.

Check the file has been renamed to "I_am_a_file2.txt".

Explanation

A subroutine, or "sub", is like a function without any return value. You can use it to do things, like rename a file, and you can still pass it arguments.

As with functions, you can leave a subroutine early. Just use "Exit Sub".

Exercise 40

Make this sub take two arguments: existingName and newName. You’ll need to separate the argument names with commas in the function header, and enclose the argument list in brackets, as with functions.

Call the sub, using the filename strings as arguments.

Note: when passing multiple arguments to a sub, you must either:


	
use the CALL keyword in front of the sub’s name, or



	
not use brackets around the argument list





For example:

CALL RenameMyFile ("I am a file2.txt", "I_am_a_file2.txt")

Or:

RenameMyFile "I am a file2.txt", "I_am_a_file2.txt"


Answer 40

Edit the script to:

Sub RenameMyFile( existingName, newName )

Dim fso

Set fso = WScript.CreateObject("Scripting.FileSystemObject")

fso.MoveFile existingName, newName

End Sub

RenameMyFile "I am a file2.txt", "I_am_a_file2.txt"

Exercise 41

Make the sub take one argument: existingName. Make it replace the spaces with underscores when renaming the file.


Answer 41

Edit the script to:

Sub RenameMyFile( existingName)

Dim fso

Set fso = WScript.CreateObject("Scripting.FileSystemObject")

fso.MoveFile existingName, Replace(existingName, " ", "_")

End Sub

RenameMyFile "I am a file2.txt"


11.3 – Process files in a folder

Example

Create a subfolder in your current folder. Call it "MySubFolder". Copy your "I am a file" files into it. If they have underscores in the file names, edit those back to spaces.

In your script, keep your subroutine for renaming files. After it, type:

Dim folderName : folderName = "MySubFolder"

Dim fso : Set fso = CreateObject("Scripting.FileSystemObject")

Dim files : Set files = fso.GetFolder(folderName).Files

Dim file

For Each file In files

MsgBox file.Name

Next

Explanation

This script displays the names of all files in a folder.

The "GetFolder" method returns a folder object. The folder object’s “Files” property contains a collection of file objects, which we can loop through.

Exercise 42

Make the script rename the files instead. You’ll need to include the subfolder name in the string you pass to "RenameMyFile".


Answer 42

As below:

Dim folderName : folderName = "MySubFolder"

Dim fso : Set fso = CreateObject("Scripting.FileSystemObject")

Dim files : Set files = fso.GetFolder(folderName).Files

Dim file

For Each file In files

RenameMyFile folderName & "\" & file.Name

Next

Exercise 43

This script is inefficient. It creates a file system object to get the files, then, inside the function, it creates a new file system object each time it renames a file. Can you make it use the same object for everything? You may need to alter the subroutine.


Answer 43

This works:

Sub RenameMyFile( existingName, fso )

fso.MoveFile existingName, Replace(existingName, " ", "_")

End Sub

Dim folderName : folderName = "MySubFolder"

Dim fso : Set fso = CreateObject("Scripting.FileSystemObject")

Dim files : Set files = fso.GetFolder(folderName).Files

Dim file

For Each file In files

RenameMyFile folderName & "\" & file.Name, fso

Next


12 – Appendices


12.1 – Ways to write and call subs and functions

Options

There are several ways of writing and calling subs and functions. Some involve brackets, or the "call" keyword, both, or neither. You’ll have seen several of them in this book. Here’s a summary.

When writing a sub or function:


	
if using arguments, use brackets.



	
if not, brackets are optional.





When calling a function:


	
if using arguments and assigning the return value, use brackets.



	
if not, brackets are optional.





When calling a sub:


	
if using arguments, use brackets and the CALL keyword or don’t use either.



	
if not, brackets and CALL are optional (and usable independently of each other).





To simplify:


	
you can just always use brackets, as long as, when calling subs, you use the CALL keyword.





Efficiency

Often, you won’t know whether something is a function or sub. You may not want to spend time checking.

In these cases, you can:


	
use brackets in all your method calls



	
run your script



	
see if you get an error message like this: "Line: 21, Char: 18, Error: Cannot use parentheses when calling a sub"



	
if you do, find that method on line 21, which you called like this: "SomeMethod(1,2,3)"



	
put "Call" in front of it: "Call SomeMethod(1,2,3)"



	
go to step 2 and repeat, until you don’t get errors.





Subs

You may see this:

SomeSub SomeArgument, SomeOtherArgument

You can pass arguments to subs without using brackets. However, as stated, if you use brackets around the argument list, you must use call:

Call SomeSub (SomeArgument, SomeOtherArgument)

There is an apparent
 exception to this rule though.

Subs behaving strangely

You may see something like this:

SomeSub (SomeArgument)

There is no CALL keyword here, yet it doesn’t error. This is because the brackets are around one argument, rather than around a list of arguments. VBScript evaluates the brackets contents like an operation.

Suppose we write:

SomeSub (2 + 2)

It evaluates to:

SomeSub 4

The final value of whatever was inside the brackets goes into the sub.

If we write:

SomeSub (4)

It evaluates to:

SomeSub 4

There are no operations inside this expression, so the value is simply copied and re-used.

If we used a variable, its value would be copied and re-used too. It’s the same principle behind using brackets to pass a variable to a method by value.

Why use CALL

Assume we already created a variable:

Dim someValue : someValue = 1

Look at this example:

SomeMethod (someValue)

Imagine the method is a sub which takes its argument by reference. It changes the value of any variable you pass it. As you used brackets, you only passed the value in, and your original variable didn’t change. You don’t realize, you re-use the variable, and you get an error.

Contrast with this way:

Call SomeMethod (someValue)

With the CALL keyword, the brackets are taken as being around the argument list, even if it’s a list of one. This means they don’t get used to re-evaluate the argument’s value. It’s the same result as if we hadn’t used brackets at all, like this:

SomeMethod someValue

It allows the method to pass by reference, as normal for the sub, and so change the variable’s value, as expected.

Finally, look at this:

Call SomeMethod ((someValue))

The extra brackets show that we are passing by value. Having to type them makes it unlikely we will pass by value without realizing it.

Using CALL with subs maintains consistency (with functions) in how the number of brackets used (one pair or two) affects variables passed as arguments.


12.2 – Other useful language features

Database Connections

VBScript supports database connections via Microsoft’s ADO, using the "ADODB.connection" object. This lets you run database queries and return the results into a "RecordSet" object.

Regex

VBScript supports regular expressions (regex) via its own, built-in, "RegExp" object. Regex lets you find patterns in text, and process strings in more complex ways.

Email

VBScript can send email, via Microsoft Outlook, using the "CDO.Message" object.

Web development

VBScript is the server-side language behind classic ASP websites. These are still in use, even in the 2020’s. Knowing VBScript lets you work on them.

Error handling

VBScript supports this, in the form of the "On Error" statement.

Support for Object-Oriented Programming

This course uses various objects, and their methods and properties. If you want to, you can write classes: templates from which you can create objects, of your own design. This allows you to organise your own code in the same way. VBScript supports this but, unlike some languages, doesn’t make you do it.

Other

This book is intended to focus on the most useful and fundamental parts of the language. However, there are many more objects, methods, and other features in VBScript. Many of the methods we covered have optional parameters which let them do even more. Keep learning! Find out what else you can do!


12.3 – Author’s Note

I hope you found this book a good way to get started with VB Script. If you did, please leave me a good review. Then more people will buy the book, and I’ll be encouraged to write others like it, which you may find useful.

If anything about the book didn’t live up to your expectations, you can let me know at:


authordarmstrong@gmail.com

